


SAM MOUSE TURBO INTERFACE

(velesoft.speccy.cz)

SAM MOUSE TURBO je nový klon myšího interface pro počítač SAM COUPE. Originál v roce 1991 vyráběla SAMco / SAMtek a prodejní cena u WEST COAST COMPUTERS byla 39.99 liber. Původní zapojení interfejsu bylo dost složité. V devadesátých letech se dostal funkční vzorek originálu do pražského klubu Kapsa, kde schema mapoval ing. Rychard Haramule. Nyní si přiblížíme zapojení a funkci původní samácké myši. Na zadním panelu počítače vedle tlačítka reset je umístěný konektor DIN 8 určený pro připojení myši. Na obrázku je popis signálů konektoru.

PIN	SIGNÁL
1	DOWN
2	UP
3	CTRL
4	LEFT
5	RIGHT
6	MOUSE INTERRUPT
7	RDMSEL
8	+5V
GND	(tělo konektoru)



Signály DOWN, UP, LEFT, RIGHT a CTRL jsou zapojeny na vodiče kurzorových klávesy počítače. SAM MOUSE interface se připojuje do myšího konektoru, z druhé strany je konektor canon 9 pro ATARI myš. Interface komunikuje se SAMem jen pomocí signálů DOWN, UP, CTRL, RIGHT a READMS. Přitom piny LEFT ani MOUSE INTERRUPT nejsou vůbec zapojeny. Myš negeneruje žádné přerušení, jen na signály DOWN, UP, CTRL a RIGHT posílá stav souřadnic a tlačítek myši. Při čtení z portu #FFFE SAM aktivuje signál RDMSEL (aktivní v log.0). Myš postupně vrací 4 bitové hodnoty souřadnic Y,X a stavy tlačítek. Y a X jsou relativní souřadnice, udávající odchylku od posledního čtení myši. Pro názornost si ukážeme softwarovou obsluhu.

NOVÝ SAM MOUSE TURBO INTERFACE



Následující rutina postupně načte a uloží stav osy Y,X a tlačítek myši.

```
start    di                ;Při čtení myši nejlépe zakážeme přerušení.
         ld hl,(#9500)      ;Adresa, kam budeme ukládat hodnoty z myši.
         ld bc,#FFFE        ;Port myši.
first_in in a,(c)           ;První dvě čtení je myš neaktivní
         ld (hl),a          ;a z portu #FFFE načítáme stav klávesnice(kurzory a cntrl)
         inc hl
         in a,(c)           ;ještě jednou čtení klávesnice
         ld (hl),a
         inc hl
         ld e,7             ;Následuje 7x čtení myši
loop     in a,(c)
         ld (hl),a          ;Všechny hodnoty postupně uložíme
         inc hl             ;za sebou
         dec e
         jr nz, loop
         ei
         ret
```

Nyní se podíváme, jaké hodnoty se z myši načetly. Od adresy #9500 je uloženo:

#9500- stav portu klávesnice (kurzory)
#9501- stav portu klávesnice (kurzory)
#9502- tlačítka myši (stisk=log.0): d0=levé, d1=pravé, d2=prostřední
#9503- horní 4 bity binárního čítače osy Y
#9504- prostřední 4 bity binárního čítače osy Y
#9505- spodní 4 bity binárního čítače osy Y
#9506- horní 4 bity binárního čítače osy X
#9507- prostřední 4 bity binárního čítače osy X
#9508- spodní 4 bity binárního čítače osy X

Zbývá snad už jen dodat, že na načtení všech devíti portů máme něco kolem 100ys od prvního načtení portu #FFFE (návěští „first_in“). Pak myš sama resetuje (nuluje) čítač pořadí čtení portů. Po devátém načtení portu #FFFE jsou zresetovány osy X a Y, takže příště vrací relativní odchylky od posledního resetu. Pokud s myší nebudeme pohybovat, měly by být od #9500 ukládány hodnoty #FF,#FF,#FF,#F0,#F0,#F0,#F0,#F0,#F0. Při stisku levého tlačítka myši vrací: #FF,#FF,#FE,#F0,#F0,#F0,#F0,#F0,#F0. Pokud se místo hodnot #F0 bude někde ukládat #FF, je jasné, že jsme nestihly včas načíst všech 9 portů #FFFE a myš již resetovala čítač pořadí portů. V takovém případě to může vypadat třeba jako: #FF,#FF,#tlačítka,#00,#00,#FF,#tlačítka,#00. O resetování osy X a Y se musíme starat sami přečtením portu #FFFE alespoň 9x během 100ys. Pak myš příště vrátí jen relativní souřadnice. Když budeme během 100ys načítat méně než 9 portů, myš nestihne zresetovat osy X,Y a ty budou vracet absolutní souřadnice. Pokud během 100ys stihneme načíst více než 9x port myši, bude myš opakovaně vracet na D0-D3 stav klávesnice(jen 1x), tlačítka, #F0, #F0, #F0, #F0, #F0, #F0.

ORIGINÁL SAM MOUSE INTERFACE



OVLADAČ PRO SAM MOUSE

Na další stránce najdete kompletní ovladač pro SAM MOUSE interface. Pro osy X a Y si můžete nastavit libovolný rozsah. MAXX a MINX nastavuje maximum a minimum v ose X. Pro osu Y platí to samé. Ovladač vrací jen 8 bitové hodnoty, takže maximum pro jakoukoli osu může být až 255. Ovladač se volá od adresy START a maximálně jednou za 100ys(nejlépe pod přerušením). Před návratem je obnoven stav přerušení (EI / DI), SP a všechny použité registry. Stačí jen vyčíst nové souřadnice a stav tlačítek:

(COORD) = nová osa X
(COORD+1) = nová osa Y
(BUTTONS) = stav tlačítek – stisknuté vrací log.1
(D0=pravé,D1=levé,D2=prostřední)

;SAM MOUSE DRIVER by VELESOFT

```

MAXX      EQU    255 ;maximum v ose X
MINX      EQU    0  ;minimum v ose X
MAXY      EQU    191 ;maximum v ose Y
MINY      EQU    0  ;minimum v ose Y

START     PUSH    AF
          LD      A,R ;uchova stav DI/EI
          DI
          LD      (SPACE),SP
          LD      SP,STACK ;vlastni SP
          PUSH    AF ;zaloha registru
          PUSH    BC
          PUSH    DE
          PUSH    HL
          LD      BC,#FFFE ;port mysi
          IN      A,(C) ;2x vynecha
          IN      A,(C) ;stav klavesnice
          IN      D,(C) ;D=tlacitka mysi
          IN      A,(C) ;vynech high Y
          IN      A,(C) ;spodni 4 bity Y
          AND     15 ;nech jen 4-bit citac
          RLCA    ;a odroluj na sve misto
          RLCA
          RLCA
          RLCA
          LD      H,A ;H=horni 4-bity Y
          IN      A,(C) ;spodni 4 bity Y
          AND     15 ;nech jen 4-bit citac
          OR      H ;A=odchylka Y (8-bit)
          CPL     ;a nakonec prevratime smer
          INC     A ;odchylky Y(0=nahore)
          LD      H,A ;H=odchylka Y (8-bit)
          LD      (ODCH+1),A ;uloz od.Y
          IN      A,(C) ;vynech high X
          IN      A,(C) ;A=vyssi 4 bity X
          AND     15 ;nech jen 4-bit citac
          RLCA    ;a odroluj na sve misto
          RLCA
          RLCA
          RLCA
          LD      L,A ;L=vyssi 4 bity X
          IN      A,(C) ;spodni 4 bity X
          AND     15 ;nech jen 4-bit citac
          OR      L ;A=odchylka X(0=vlevo)
          LD      L,A ;L=odchylka X (8-bit)
          LD      (ODCH+0),A ;uloz od.X
OSA_X     LD      A,(COORD+0) ;osa X
          ADD     A,L ;
          BIT     7,L ;test smeru odchylky
          JR      Z,RIGHT
LEFT      JR      C,NXTLEFT
MNX       LD      A,MINX ;minimum osy X
          JR      ENDX ;
NXTLEFT   CP      MINX ;test minima X
          JR      ENDX
NXTRGHT   CP      MAXX ;test maxima X

```

```

ENDX      JR      NC,MAXX
          LD      (COORD+0),A ;nove X

;pro pocitani osy Y platí to same co pro osu X
OSA_Y     LD      A,(COORD+1)
          ADD     A,H
          BIT     7,H
          JR      Z,UP
DOWN      JR      C,NXTDOWN
MNY       LD      A,MINY
          JR      ENDY
UP        JR      NC,NEXTUP
MXY       LD      A,MAXY
          JR      ENDY
NEXTUP    CP      MAXY
          JR      NC,MXY
          JR      ENDY
NXTDOWN   CP      MINY
          JR      C,MNY
          LD      (COORD+1),A ;nove Y

;test tlacitek mysi
          LD      A,D ;A=stav tlacitek
          CPL     ;pro lepsi testovani neguje
          AND     7 ;a necha jen tlacitka
          LD      D,A ;vysledek do D
          RLCA    ;nasleduje usporadani
          BIT     3,A ;bitu stavu tlacitek
          JR      Z,NIC ;podle vzoru
          OR      1 ;Kempston mouse ze ZX
          AND     7 ;Spectra
          LD      (BUTTONS),A ;uloz

          POP     HL ;obnoveni registru
          POP     DE
          POP     BC
          POP     AF
          LD      SP,(SPACE) ;obnov SP
          JP      PO,IM_DI ;testuj EI/DI
IM_EI     POP     AF ;navrat s povolenym
          EI      ;prerusenim
          RET
IM_DI     POP     AF ;navrat se zakazanym
          RET     ;prerusenim


COORD     DEFB    MINX,MINY ;osy X,Y
ODCH      DEFB    0,0 ;odchylky X,Y
BUTTONS   DEFB    0 ;stav tlacitek
SPACE     DEFW    0 ;misto pro zalohu SP
DEFS      16 ;misto pro vlastni
STACK     ;zasobnik ovladace mysi

```

Nové souřadnice myši nastavíte zápisem na adresu (COORD) a (COORD+1).

Schema originál SAM MOUSE interface (v2.1) (mapoval Rychard Haramule)


100% testováno – souhlasí se zapojením originál SAM MOUSE



Schema SAM MOUSE interface z technického manuálu k SAM COUPE


! POZOR – v tomto schématu nesouhlasí pinout a zapojení !

Nejspíš se jedná o vývojové schema





Dosud jsme si popisovali jen originál interface. Ten se již nevyrábí, a velice těžko se dá sehnat. Na www.samcoupe.com máte sice možnost objednat novou verzi sam mouse interface (autor: Colin Piggot), ale cena je kolem 50 liber a navíc interface pracuje se sériovou myší, kterou dnes už v obchodě nekoupíte. Postavil jsem tedy nový myší interface, který je 100% kompatibilní s originálem. Kromě kompatibility obsahuje i několik rozšířených funkcí, jako například čtyřnásobně rychlejší pohyb s možností zpomalení na původní rychlost. Dále je tu možnost prohození funkce levého a pravého tlačítka myši. Do interface se již zapojuje běžná PS/2 mouse. Procesor PIC se stará o konverzi PS/2 na signály AMIGA mouse a CPLD funkčně zastupuje původní interface. Bližší informace najdete na velesoft.speccy.cz. Nyní si popíšeme jednotlivé části na desce, kterou si můžete prohlédnout na následujícím obrázku. Ve velké čtvercové patici uprostřed desky je obvod CPLD nahrazující funkci originál sam mouse, vpravo je obvod PIC. Vlevo dole je přepínač ON/OFF mouse. Tlačítko vpravo dole při stisku umožňuje měnit rozšířené funkce.

SAM MOUSE TURBO INTERFACE



Pinout konektoru pro DIN8





ROZŠÍŘENÉ FUNKCE

Kdykoli za provozu můžete měnit nastavení myši. Stačí přidržet stisknuté tlačítko EXTRA(vpravo dole) nebo 4./5. tlačítko u vícetlačítkových myší, a k tomu stisknout prostřední nebo pravé. Prostředním tlačítkem se přepíná rychlost myši (originál / turbo 4x), pravé tlačítko prohodí funkci levého a pravého tlačítka.


PŘIPOJENÍ JOYSTICKU

SAM MOUSE TURBO interface umožňuje připojit joystick s konektorem CANON 9. Obvod CPLD se sám stará o to, aby joystick nevadil ve funkci myši. Myš je u SAMa bohužel připojená na vodiče klávesnice a sdílí stejný port s kurzorovými klávesami. Ve skutečnosti na SAMovi není možné současně používat myš a kurzory. Joystick připojený do myšího interface nyní funguje jako kurzory klávesnice+CNTRL a je čitelný na portech #FFFE jen v době, kdy myš není čitelná (první dvě čtení portu, když čteme stav klávesnice). Tímto způsobem je zajištěno, že bez omezení funguje myš i joystick(kurzor). Některé hry používají ovládání myši i kurzorem současně, např. Legend Of Eshan.

PS/2 nebo USB

Interface je možné osadit konektorem PS/2 nebo USB. Veškerou obsluhu připojené myši obstarává obvod PIC. Ten ovšem pracuje jen s PS/2 protokolem. Takže připojená PS/2 myš by měla vždy pracovat bez problémů, zatímco USB myš musí umět PS/2 protokol. Nejjistější je koupit USB myš dodávanou s redukcí USB > PS/2, kde je kompatibilita zaručena.

Spodní strana desky – pohled na konektor PS/2(USB)



Na plošném spoji pod konektorem pro myš jsou barevně označené 3 vodiče. Při osazení konektoru PS/2 je třeba přeškrábnout 2 vodiče v místě žlutých značek. V případě USB stačí přerušit jediný vodič v místě červené značky.

Zdrojový kód CPLD (ABEL)

MODULE SAMMOUSE

```
// 11.12.2005
// Zdrojovy kod v jazyce ABEL pro CPLD XILINX XC9572XL-PC44

// SAM MOUSE TURBO interface for SAM COUPE

// CPLD funguje jako konvertor signalu Amiga mouse na SAM mouse.
// SAM mouse je mys pouzivana u pocitace SAM COUPE.

// Popis funkce:
// SAM mouse je citelna na portu #FFFE
// Podle poradí čtení tohoto portu vrácí mys různé údaje:
// Při 1. a 2. nactení portu ponechá na D0-D3 stav klávesnice.
// (připojený joystick, emuluje kurzory a CTRL)
// 3. - stav tlačítek(D0=levé,D1=prostřední,D2=právé,D3=vždy log.1)
//      (stisknutí tlačítka vrácí log.0)
// 4. - nejvyšší 4.bit čísla osy Y
// 5. - střední 4.bit čísla osy Y
// 6. - nejnižší 4.bit čísla osy Y
// 7. - nejvyšší 4.bit čísla osy X
// 8. - střední 4.bit čísla osy X
// 9. - nejnižší 4.bit čísla osy X
// Další čtení se opakuje od bodu 2, ale čítá osu X a Y dále vrácí
// jen hodnotu 0. Mys si nuluje čítání obou os vždy po 9. nactení
// nactení portu. Ovladač myši musí být napsán tak, aby stihl během
// 100ms nacist a někam uložit postupně všech 9 portů! Pokud toto
// nedodrží, mys nebude resetována a bude vždy vracet jen absolutní
// souřadnice. Jinak čítá vrácí relativní souřadnice, tedy odchylku od
// poslední nactené souřadnice X,Y. Max. odchylka může být -2048 až
// +2047(pro každou osu je použit 12-ti bitový binární čítací)

//CPLD obsahuje pár vylepsení oproti original SAM MOUSE:
// 1) čtyřnásobné zvýšení rychlosti pohybu myši s možností přepnutí na "original"
// 2) možnost prohození funkce LEVEHO a PRAVEHO tlačítka myši.
// 3) prepínacem ON/OFF lze mys vypnout/zapnout

//Pro použití obou funkcí je nutné přidržet stisknutí tlačítka EXTRA(nebo 4./5.tlačítka
//myši) a současně stisknout prostřední tlačítka(prepinání rychlosti) nebo právé
//tlačítka(prohození funkce L a R tlačítek). Při držení EXTRA nebo 4./5.tlačítka je
//automaticky blokováno čtení stavu tlačítek myši. Software tak nikdy nepozná, že
//uživatel přepíná nastavení rychlosti nebo tlačítek myši.

//vypínacem ON/OFF se zapíná/vypíná mys. Při ON je připojený joystick citelný
//jen na stavu kurzorových kláves a nekoliduje s myši. Při OFF je mys neaktivní
//a joystick je na portu #FFFE citelný vždy.

////////////////////////////////////

//VSTUPNÍ PINY

//registry s předchozím stavem clonek myši v ose X
BXA    NODE istype 'reg';
BXB    NODE istype 'reg';
BX = [BXA,BXB];

//registry s předchozím stavem clonek myši v ose Y
BYA    NODE istype 'reg';
BYB    NODE istype 'reg';
BY = [BYA,BYB];

//signály směru při pohybu myši
LEFT   NODE istype 'COM';
RIGHT  NODE istype 'COM';
UP      NODE istype 'COM';
DOWN   NODE istype 'COM';

//řídící signály z CPU
RDMSEL  PIN 27;//CLK čítá portu zvyšuje pořadí čtení
ENCOUNT PIN 19;//log.0 = ENABLE čítá X,Y + LOAD čítá pořadí čtení
```



```

//signaly z A-MOUSE(PICu)
XA      PIN 1; //joy DOWN
XB      PIN 43; //joy RIGHT
YA      PIN 40; //joy UP
YB      PIN 42; //joy LEFT
RIGHTB PIN 6; //joy FIRE 2
LEFTB   PIN 4; //joy FIRE 1
MIDDLEB PIN 7; //joy FIRE 3

//VOLNE PINY S PULL-UP
MOUSELED PIN 39 ISTYPE 'REG'; //ledka detekuje cteni softwarove cteni mysi
//X2      PIN 38; //NIC
JOYUP    PIN 37; //UP
JOYFIRE  PIN 36; //FIRE
JOYDOWN  PIN 35; //DOWN
JOYLEFT  PIN 34; //LEFT
JOYRIGHT PIN 33; //RIGHT
//X8      PIN 9; //NIC
//X9      PIN 12; //NIC

//VYSTUPNI PINY
D3       PIN 20 istype 'COM'; //LEFT
D2       PIN 26 istype 'COM'; //DOWN
D1       PIN 24 istype 'COM'; //UP
D0       PIN 25 istype 'COM'; //CNTRL
D4       PIN 22 istype 'COM'; //RIGHT
//MSEINT  PIN 18 istype 'com';

//INTERNI VYSTUPY
XQ11..XQ0 NODE istype 'reg';
XCOUNTER = [XQ11..XQ0]; //citac pro osu X
YQ11..YQ0 NODE istype 'reg';
YCOUNTER = [YQ11..YQ0]; //citac pro osu Y

R3..R0    NODE istype 'reg'; // citac poradí ctení portu
RCOUNTER = [R3..R0];

//=====
// FEATURES ///

EN_MOUSE  PIN 28; //BLOKUJE MYS(prepinac sepnuty v log.0)
EXTRA     PIN 3; //pri stisku je zapnuty extra mod
BUTTON4   PIN 44; //4.tlacitko mysi
BUTTON5   PIN 8; //5.tlacitko mysi
EXBUT     NODE istype 'reg'; //reg.stav layoutu tlacitek(EXBUT/NORMAL)
SLOW      NODE ISTYPE 'reg'; //reg.rychlosti pohybu SLOW / FAST 4x

////////////////////////////////////
EQUATIONS

//smery pohybu mysi
RIGHT = !BXA & !BxB & XA & !XB
# BXA & !BxB & XA & XB & !SLOW //R,L,U,D = ve SLOW rezimu 3x ze 4 stavu neaktivni
# BXA & BxB & !XA & XB & !SLOW
# !BXA & BxB & !XA & !XB & !SLOW;

LEFT = !BXA & !BxB & !XA & XB
# !BXA & BxB & XA & XB & !SLOW
# BXA & BxB & XA & !XB & !SLOW
# BXA & !BxB & !XA & !XB & !SLOW;

DOWN = !BYA & !BYB & YA & !YB
# BYA & !BYB & YA & YB & !SLOW
# BYA & BYB & !YA & YB & !SLOW
# !BYA & BYB & !YA & !YB & !SLOW;

UP = !BYA & !BYB & !YA & YB
# !BYA & BYB & YA & YB & !SLOW
# BYA & BYB & YA & !YB & !SLOW
# BYA & !BYB & !YA & !YB & !SLOW;

//pri kazde zmene clonek uchova aktualni stav clonek mysi do internich registru.
BXA := XA;
BxB := XB;
BX.clk = !BXA & !BxB & XA & !XB

```

```

# BXA & !BXB & XA & XB
# BXA & BXB & !XA & XB
# !BXA & BXB & !XA & !XB
# !BXA & !BXB & !XA & XB
# !BXA & BXB & XA & XB
# BXA & BXB & XA & !XB
# BXA & !BXB & !XA & !XB;

BYA := YA;
BYB := YB;
BY.clk = !BYA & !BYB & YA & !YB
# BYA & !BYB & YA & YB
# BYA & BYB & !YA & YB
# !BYA & BYB & !YA & !YB
# !BYA & !BYB & !YA & YB
# !BYA & BYB & YA & YB
# BYA & BYB & YA & !YB
# BYA & !BYB & !YA & !YB;

//clock pro citace os X a Y
XCOUNTER.clk = LEFT & !RIGHT & !ENCOUNT
# RIGHT & !LEFT & !ENCOUNT;
XCOUNTER.ar = R3;//reset

YCOUNTER.clk = UP & !DOWN & !ENCOUNT
# DOWN & !UP & !ENCOUNT;
YCOUNTER.ar = R3;//reset

//citac pro osu X
WHEN (!LEFT & RIGHT & !ENCOUNT) //pohyb vpravo
THEN XCOUNTER := XCOUNTER + 1
WHEN (!RIGHT & LEFT & !ENCOUNT) //pohyb vlevo
THEN XCOUNTER := XCOUNTER - 1

//citac pro osu Y
WHEN (!DOWN & UP & !ENCOUNT) //pohyb nahoru
THEN YCOUNTER := YCOUNTER + 1
WHEN (!UP & DOWN & !ENCOUNT) //pohyb dolu
THEN YCOUNTER := YCOUNTER - 1

//citac poradí cteneho portu #FFFE. Zvyšuje se po ukončení čtení portu #FFFE.
RCOUNTER.CLK = RDMSSEL;//citac poradí portu se zvýší vždy po načtení portu.
RCOUNTER := RCOUNTER+1;
RCOUNTER.AR = !ENCOUNT//citac portu resetovan na konci práci s myši-100ys od 1.čtení portu
# EN_MOUSE;//a také pokud není sepnutý prepínac EN_MOUSE(vypne mys)

!MOUSELED = R0// LEDka svítí když je softwarově obsluhována mys.
# R1
# R2
# R3;

//output enable pro datové signály(propouští pouze data v log.0, aby fungovala i klávesnice)
D0.oe = !D0 & !RDMSSEL;
D1.oe = !D1 & !RDMSSEL;
D2.oe = !D2 & !RDMSSEL;
D3.oe = !D3 & !RDMSSEL;
D4.oe = !D4 & !RDMSSEL & !R2 & !R1 & !R0;

//multiplex dat. Prepína 8x port SAM-MOUSE
D0 = !R2 & !R1 & !R0 & !JOYFIRE//první dvě čtení portu #FFFE vrací stav joysticku, který tak
# !R2 & !R1 & R0 & LEFTB & !EXBUT//3.čtení = stav tlačítek - D0=LEVE,D1=PROSTREDNI,
// D2=PRAVE,D3=vždy log.1)
# !R2 & !R1 & R0 & RIGHTB & EXBUT//prohozené levo a prave tlačítka
# !R2 & !R1 & R0 & !BUTTON4 //při stisku 4.tlačítka není citelný stav všech tlačítek
# !R2 & !R1 & R0 & !BUTTON5 //při stisku 5.tlačítka není citelný stav všech tlačítek
# !R2 & R1 & !R0 & YQ8 //nejvyšší 4.bit citac osy Y
# !R2 & R1 & R0 & YQ4 // střední 4.bit citac osy Y
# R2 & !R1 & !R0 & YQ0 //nejnižší 4.bit citac osy Y
# R2 & !R1 & R0 & XQ8 //nejvyšší 4.bit citac osy X
# R2 & R1 & !R0 & XQ4 // střední 4.bit citac osy X
# R2 & R1 & R0 & XQ0; //nejnižší 4.bit citac osy X

D1 = !R2 & !R1 & !R0 & !JOYUP

```

```

# !R2 &!R1 & R0 & MIDDLEB
# !R2 &!R1 & R0 & !BUTTON4
# !R2 &!R1 & R0 & !BUTTON5
# !R2 & R1 &!R0 & YQ9
# !R2 & R1 & R0 & YQ5
# R2 &!R1 &!R0 & YQ1
# R2 &!R1 & R0 & XQ9
# R2 & R1 &!R0 & XQ5
# R2 & R1 & R0 & XQ1;

D2 = !R2 &!R1 &!R0 & !JOYDOWN
# !R2 &!R1 & R0 & RIGHTB &!EXBUT
# !R2 &!R1 & R0 & LEFTB & EXBUT
# !R2 &!R1 & R0 & !BUTTON4
# !R2 &!R1 & R0 & !BUTTON5
# !R2 & R1 &!R0 & YQ10
# !R2 & R1 & R0 & YQ6
# R2 &!R1 &!R0 & YQ2
# R2 &!R1 & R0 & XQ10
# R2 & R1 &!R0 & XQ6
# R2 & R1 & R0 & XQ2;

D3 = !R2 &!R1 &!R0 & !JOYLEFT
# !R2 &!R1 & R0
# !R2 & R1 &!R0 & YQ11
# !R2 & R1 & R0 & YQ7
# R2 &!R1 &!R0 & YQ3
# R2 &!R1 & R0 & XQ11
# R2 & R1 &!R0 & XQ7
# R2 & R1 & R0 & XQ3;

D4 = !JOYRIGHT;

```

//PRIDANE FEATURES

//BUTTON4/5 + prave tlacitko = prohodi funkci leveho a praveho tlacitka

```

EXBUT.clk = !RIGHTB & !BUTTON4
# !RIGHTB & !BUTTON5
# !RIGHTB & !EXTRA; //nebo pri drzeni tlacitka EXTRA
EXBUT      := !EXBUT;

```

//BUTTON4/5 + prostredni tlacitko = meni rychlost pohybu SLOW(original) / TURBO 4x

```

SLOW.CLK   = !MIDDLEB & !BUTTON4
# !MIDDLEB & !BUTTON5
# !MIDDLEB & !EXTRA;
SLOW       := !SLOW;

```

END